1. Welcome to the *GrapeVine* tutorial!

GrapeVine is a tool for graph transformations. ***GrapeVine*** can be used "under the hood" as a library in software projects, but it also comes with a computational notebook UI, which leverages Gorilla.

A computational notebook consists of mulltiple *worksheets* (you are looking at one now). Each worksheet consists of *static* and *executable segments*.

1.1 Static segments

Static segments can contain text using markdown, html, ***LaTeX***, images and fancy things like Mermaid diagrams that can be edited online. See below for an example ER-diagram that shows the ***GrapeVine*** meta model (which you don't need to understand right now). If you want to edit the diagram, right-click opens in a new window with Mermaid live).



Fig. 1. ***GrapeVine*** Meta Model in Entity-Relationship Notation

1.2 Executable segments

Executable code is written in Clojure. (***GrapeVine*** code is a domain specific language extension of Clojure).

Executable segments are not automatically invoked when a worksheet is loaded.

Shift + enter evaluates a single code segment.

Ctrl + Shift + enter evaluates all code segments in a worksheet.

```
1 (+ 2 3)
```

```
5
```

1.3 Commands

Press ctrl+g twice in quick succession or click the menu icon (upper-right corner) for an overview on commands.

1.4 Using *GrapeVine* - Resources and namespaces

To make ***GrapeVine*** available, *its library must be imported.* Under the hood, ***GrapeVine*** connects to the graph database NEO4J. The simplest way to do the import is by using a `(use 'grape.core')` form, but it's best practice to define a separate namespace for each worksheet. Below creates a worksheet of name `tutorial`.

(**Remember: The statement below must be executed before any *GrapeVine*-related commands.** (Shift-enter) Establishing the session may take a few seconds. Wait until the green light is gone.)

Extend the statement below to import from other worksheets, if your project uses multiple worksheets.

```
1 (ns tutorial
2   (:require [grape.core :refer :all]))
```

```
nil
```

2. Graphs and graph persistence

GrapeVine uses a database to store graphs. Graphs, once defined are persisted. Arbitrarily many graphs can be created. Morevover, graphs are maintained in a [fully persistent data structures](), i.e., *all* versions of a graph can be accessed and modified.

A new (empty) graph is created using the `(newgrape)` function:

```
1 (newgrape)
```

```
("79c0ff01-77d1-4047-bed1-bff3feabd6bd")
```

As shown above, graphs are internally identified with a unique ID. That may not be convenient for the worksheet, but we can always define *vars* to refer to graphs. The statement below defines a *var* that refers to an empty graph.

> *Note: if you are not familiar with Clojure, you can think of vars are similar to variables in other languages.*

```
1 (def g0 (newgrape))
```

```
#'tutorial/g0
```

The built-in form `view` can be used to generate an image of the content of a graph. (These images are saved when the worksheet is saved.)

```
1 (view g0)
```



Unsurprisingly, the graph is empty. Perhaps suprisingly, it will always remain so. (Graphs are immutable in ***GrapeVine***).

2.1 Graph Enumerations (*Grapes*)

If you are very perceptive (and have some knowlege of Clojure data types), you may have noted that the above call to `newgrape` returned a *sequence* (with a single element) rather than a simple string (as indicated by the brackets `()`). Such a sequence is called a **GRAPH set Enumeration** or ***grape*** for short.

In ***GrapeVine***, all operations that produce graphs actually return *grapes* rather than graphs, even those ones that are guaranteed to generate a single graph only, such as the `newgrape` function.

Using a single data type for operations that can produce graphs facilitates composability of operations (as we will see later).

This technique of using a single data type is well known also from other domains, such as the data type of a "relation" in relational databases. (Even queries that are guaranteed to produce a single, unique result will still generate a relation with a single element.)

For simplicity and if the context is clear, we may continue to refer to ***grapes*** with a single element as "graphs".

3. Graph transformation rules

Graph transformation rules ("rules" for short) are defined in *hybrid* language (textual input with graphical view). There are three example rules below.

1. The first rule (**`hello`**) creates a new node of type *Hello*
2. The second rule (**`world`**) looks for an existing node of type `Hello` and connects it with a new node of type `World`.
3. The third rule (**`vine`**) looks for two nodes of type `Hello` and `World`, respectively, which are connected with a `to` edge and replaces the `World` node with a node of type `Grape`.

As we can see in the visualization, green colour marks new graph elements, red colour marks deleted ones, and black colour marks "preserved" elements.

> *Note: While graphs are persisted in the database, rules are not. If you want to execute any of these rules, you need to evaluate the code segment below first to declare them. Otherwise, the rules will not be known. Also, note that enclosing these rules in a `(list ...)` form has no significance and is purely cosmetic. The only reason for having it is to render multiple rules horizontally in the visualization. Feel free to delete it if you want vertical listing of output.*

```
(list

(rule hello []
   (create
     (node :Hello)))

(rule world []
   (read
      (node h:Hello))
   (create
      (node w:World)
      (edge :to h w )))

(rule vine []
   (read
      (node h:Hello)
      (node w:World)
      (edge e:to h w ))
   (delete w e)
   (create
      (node g:Grape)
      (edge :to h g)))
 )
```



4. Rule application

Applying a rule to a graph is as simple as calling a function with the rule's name with the graph as an argument. Note that this will produce a new graph (if the rule is applicable). (Graphs are immutable in ***GrapeVine***, so `g0` will always remain an empty graph.)

```
(hello g0)
```

```
("9b976167-8264-45c1-a269-968fac7e27d1")
```

Applying rule `hello` to graph `g0` worked, since it generated a new graph. If we want to visualize the resulting graph, we could copy the above result and use it as an argument for `viewgraph`, i.e., call `viewgraph '("7eff0750-abb7-428d-81ef-6c6483a17c3a")`. However, that's cumbersome. Of course, we could also just nest the functions:

```
(view (hello g0))
```



> *Note: The graph ids of the two applications of rule `hello` are differnt. Indeed we actually get a new graph each time a rule is applied.*

For convenience, ***GrapeVine*** also provides a shortcut symbol _ (underscore), to refer to the "last created graph" (resp. "last created ***grape***"). So the following also works:

```
1 (hello g0)
2 (view _)
```



*Note: The shortcut (underscore) is like a global variable that is updated as a **side effect** when a new graph is produced. It should be clear that this breaks the functional computation paradigm otherwise used in **GrapeVine**. From a practical perspective this may mean that dynamic segments in a **GrapeVine** worksheet are no longer idempotent, but need to be executed in order to achieve a repeatable result.*

Now let's see the result of an unsuccessful rule application. Clearly, rule `world` requires the presence of a `Hello` node. So the following should not work:

```
1 (world g0)
```

```
nil
```

However, rule `world` should be applicable to the graph that's produced when applying rule `hello` to graph `g0`:

```
1 (view (world (hello g0)))
```



Deeply nested function may be hard to write and read. (Too many parentheses...) Consider this:

```
1 (view (vine (world (hello (newgrape)))))
```



Clojure provides ***threading macros***, to make functional composition more readable. The following expression is equivalent to the above:

```
(-> (newgrape) hello world vine view)
```



4.1 Matching semantics

Rules may be defined to use ***homomorphic*** or ***isomorphic* matching** semantics. Homomorphic semantics is the default, and allows different graph elements in the rule to match the same graph elements in the graph that the rule is applied to. To understand this better, consider a graph (`g1`) with one `Hello` node:

```
(def g1 (-> (newgrape) hello))
(view g1)
```



Now consider the following rule. It matches "two" `Hello` nodes with a new relationship `rel`.

```
(rule rel []
   (read
      (node h:Hello)
      (node h2:Hello))
   (create
      (edge e:rel h h2)))
```



Is this rule applicable to graph `g1`? The answer to this question depends on the matching semantics. If ***homomorphic*** matches are allowed, the two `Hello` nodes in rule `rel` may in fact be matched to the *same* node in the graph the rule is applied to. This would result in a rule with a `rel` edge to itself:

```
(view (rel g1))
```



The above (homomorphic) matching semantics is the default. However, matching multiple nodes in a rule to a single node in the graph may not be desirable in some appliations. In this case, the rule designer can specify the rule to enforce ***isomorphic*** matching. The following rule `rel2` is identical to the above rule `rel`, except that it requires *isomorphic* matches.

```
(rule rel2 []
   (read :iso
       (node h:Hello)
       (node h2:Hello))
   (create
       (edge :rel h h2)))
```



Applying that rule to our graph with a single `Hello` node should fail:

```
(rel2 g1)
```

```
nil
```

Let's add another `Hello` node:

```
(def g2 (hello g1))
(view g2)
```



We would now expect that an isomorphic match for rule `rel2` can be found:

```
(view (rel2 g2))
```




```
(                                                                                )
```

4.1.1 Mixed matching semantics

At times we may want to allow homomorphic matches for parts of the rule but enforce an isomorphism constraint for another part of the rule. This can be done by specifying **application conditions**. We will talk aboout application conditions later in this tutorial. However, as a preview, here is a rule (`rel3`) that uses an application condition to ensure that two different nodes are matched. Since `rel3` only has two nodes, its semantics is the same as that of rule `rel2`:

```
(rule rel3 []
   (read
       (node h:Hello)
       (node h2:Hello)
       (condition "not h=h2"))
   (create
       (edge :rel h h2)))
```




```
(view (rel3 g1))
```

```
()
```

```
(view (rel3 g2))
```



4.2 Non-determinism during matching

Consider the above example of graph `g2` again. Is there only one possible way to apply rule `rel2`? Indeed, there are **two** possible ways to apply that rule. The roles of the two nodes could also have been reversed, which would create the relationship in the *opposite* direction.

Applying a rule to a graph non-deterministically picks one out of possibly many valid matches.

4.3 Starred rule applications

In addition to simple rule applications, which non-deterministically pick one out of possibly many matches, ***GrapeVine*** provides **starred rule applications**. A starred rule application applies a rule so that *all* possible matches are performed. This generates a *set* of graphs, where each graph represents the result of a possible rule application. The output is a ***grape***, which contains more than a single graph. (See Section 2.1)

> *Note: this is the point in the tutorial when we start talking about "**grapes**" rather than graphs when referring to the input or output of operations.*

Invoking a starred rule is simply done by adding an asterisk after the rules name:

```
(rel2* g2)
```

```
("51b0a06f-2c0c-42f3-b7c5-e97685c9a509" "fc6e97cf-c550-42fa-9972-649ac725d34f")
```

Indeed, we see that the *grape* produced by the starred application of rule `rel2` to our graph `g2` has two graphs. We can view the result by using `view`. (Yes, `view` works on *grapes* that contain more than a single graph).

```
1 (view (rel2* g2))
```



How many possible applications exist for our original (homomorphic) rule `rel`, when applied to graph `g2`? There should be four. Let's see...

```
1 (view (rel* g2))
```



4.4. Sifting *grapes* for "distinct" graphs

The above example shows a ***grape*** with four graphs. However, there are really only two *distinct* graphs in that set. In other words, there are two sets of graphs that can be seen as "equivalent" (when disregarding the node identifiers). We say that these graphs are identical up to isomorphism.

GrapeVine has the distinct operator (`dist`) operator to filter out such "duplications" (see below)

```
1 (-> g2 rel* dist view)
```



Note: Since graph isomorphism checking is expensive, the current implementation of the `dist` *operator is based a weaker notion of checking for graph bisimilarity. Two bisimilar graphs are likley isomorphic but not necessarily so. A future version of* ***GrapeVine*** *may do further isomorphism tests, similar to what has been implemented in* GROOVE.

4.5 Applying rules to *grapes* (containing multiple graphs)

So far, we only applied rules to single graphs, i.e., ***grapes*** with a single element. However, rules can also be applied to ***grapes*** that enumerate multiple graphs. Consider the ***grape*** produced by the following statement. (We create a graph with two `Hello` nodes and then carry out a starred rule application of `world`.) The ***grape*** should have two graphs:

```
1 (-> (newgrape) hello hello world* view)
```



Applying rule `vine` to the above ***grape*** will apply the rule to *each* graph in the ***grape*** and produce a new ***grape*** that enumerates all resulting graphs. (This also works for starred rule applications.)

```
(-> (newgrape) hello hello world* vine view)
```



5. History Graphs, Traces and Derivations

As explained above, rules are applied to graphs (in ***grapes***) and are used to *derive* new graphs. This history of **derivations** of graphs can themselves be represented as a graph, where the nodes represent graphs and edges represent *occurrences* of graph transformations (i.e., rule applications).

5.1 History Graph

We define the **history graph** of a given graph g as the graph that contains all graphs (and derivations) that g depends on or that depend on g.

The above definition can easily be extended to ***grapes***: Since computations with ***grapes*** always start with a ***grape*** that contains a single graph only, the history graph of a ***grape*** is defined as the history graph of the unique original start graph that all graphs in the ***grape*** depend on.

We can generate the *history graph* of a ***grape*** using the `history` function:

```
(def h (-> (newgrape) hello hello world* vine history))
```

```
#'tutorial/h
```

A history graph is just a regular graph and can be operated on accordingly. For example, we can view the history graph we just created using the view function.

```
1 (view h)
```



5.2 Derivation steps

The history shown above consists of four derivation steps:

1. Step 1: hello
2. Step 2: hello
3. Step 3: World* (two concurrent derivations)
4. Step 4: vine (two concurrent derivations)

Each of these steps can be represented by a ***grape***. We can use the form `steps` to attain this sequence of ***grapes*** that characterize a history graph:

```
1 (steps h)
```

```
(("3a788067-8e54-4c41-86a6-69acf33ec8a8") ("af6de187-4a92-4009-a0e9-404962d332b1") ("ae3c4ab7-
7287-4c5b-8746-cc088eb017cf" "7e7ad93a-57a4-4507-bef9-0467170306c9") ("6ac41796-9d6f-4640-8e43-
270812e690cf") ("7a0fded4-cf08-4e68-9307-b720d1eeca1f"))
```

We can use the form `viewsteps` to visualize the details for each derivation.

```
1 (viewsteps (steps h))
```



5.3 Traces

In many cases, we are not interested in the entire history graph of a ***grape***. Rather, we may just be interested the derivation steps that led to a particular graph. We call the (linear) sequence of derivation steps that let to a particular graph the **trace** of that graph.

Anologously to history graphs, traces can be represented in graph form (showing linear chains of derivations).

For a given ***grape***, a set of traces for all included graphs can be computed using the `traces` form. As we can see above, there are two traces in our example:

```
1 (-> (newgrape) hello hello world* vine traces view )
```



(



)

Of course, we can also view details of the derivation steps on traces:

```
1 (-> (newgrape) hello hello world* vine traces steps viewsteps )
```



6. Graph Constraints

Graph constraints define conditions on the state of graphs. Graph constraints are defined based on a notion of an *atomic graph constraint* of form **if X then C**, where X describes a pattern to be searched for in a graph and C describes a necessary extension of that pattern for all found matches.

This concept of an atomic graph constraint (*atomic constraint* for short) has been formalized as $\forall(c : X \longrightarrow C)$ (see Orejas et al. for details).

6.1 Basic constraints

An atomic constraint where $X = \emptyset$ is called a *basic atomic constraint* or *basic constraint* for short. It can be written as $\exists C$.

Consider the following basic constraint, which matches a `Hello` node that is connected to a `Grape` node via a `to` edge.

> *Note: By convention, names of constrains should end with a exclamation mark.*

```
(constraint hello-to->grape! []
   (node h:Hello)
   (node p:Grape)
   (edge :to h p))
```



6.2 Using constraints to filter *grapes*

Like rules, constraints can be applied to *grapes* and filter out those graphs that do not satisfy the constraint. Applications can be positive or negated.

Let's try:

```
(hello-to->grape! (newgrape))
```

```
()
```

As expected, the above pattern cannot be found in an empty graph. Let's apply the constraint in negated form. (i.e., we assert that the "hello-to->grape!" pattern may not exist). This is done by appending a minus sign (**-**) to the call of the constraint.

```
(hello-to->grape!- (newgrape))
```

```
("12f918ce-d728-4f46-a293-3cfadaaf64a0")
```

We we see above that the empty graph satisfies this negated constraint.

Conversely, let's generate a graph that satisfies the positive constraint (see below):

```
(-> (newgrape) hello world vine hello-to->grape!)
```

```
("049c7b39-3869-4411-a3b4-5203f6a5d8ee")
```

Observe that testing a graph constraint only returns a graph if it satisfies the constraint.

> *Graph constraints can act as filters on **grapes**. They filter out all graphs that do not satisfy the constraint.*

Let's see another example with the `rel` rule defined earlier. As previously noted, doing a starred application of that rule to a graph with two `Hello` nodes produces four possible derivations.

```
(-> (newgrape) hello hello rel* view)
```



Now let's consider the following constraint `self-related!`, which matches a node that has a relationship with itself:

```
(constraint self-related! []
        (node h)
        (edge e:rel h h ))
```



As we can see below, applying the graph constraint filters out those graphs that do not contain a "self-related" node:

```
(-> (newgrape) hello hello rel* self-related! view)
```



Conversely, applying the negated constraint exclude graphs that satisfy the constraint:

```
(-> (newgrape) hello hello rel* self-related!- view)
```



6.3 Atomic Constraints

As described above, atomic constraints have the general form **if X then C**. (Remember that basic constraints are a special case, were the pattern X is empty, intuitively: *if TRUE then* X)

Below is an example for an atomic constraint that is *not* a basic constraint. A graph satisfies this constraint if (and only if) all `Hello` nodes are connected to a `Grape` node by a `to` edge.

As we can see, the pattern that belongs to the "if" part is represented in black colour.

```
(cond-constraint if-hello-then-to-grape! []
       (IF   (node h:Hello))
       (THEN (node p:Grape)
             (edge :to h p)))
```



An empty graph satisfies this constraint:

```
(-> (newgrape) if-hello-then-to-grape!)
```

```
("8edd7717-d8aa-4c5f-afa7-85d2eddb3973")
```

The following graph does not:

```
(-> (newgrape) hello world if-hello-then-to-grape!)
```

```
()
```

But this one does:

```
(-> (newgrape) hello world vine if-hello-then-to-grape! view)
```



( )

6.4 Constraint clauses

In mathematical logic, a *clause* is a formula of the form $L_1 \vee \ldots \vee L_n$, where each L_i is a positive or negative literal.

GrapeVine provides the `constraint-clause` form for defining constraints in clausal form, where each literal is a positive or negative atomic constraint. For example, the form below defines a constraint claus (named `self-or-not-hello-to-grape!`) as a disjunction of the two atomic constraints `self-related!` and `if-hello-then-to-grape!`).

```
(constraint-clause  self-or-not-hello-to-grape!
                    [self-related! if-hello-then-to-grape!])
```

```
#'tutorial/self-or-not-hello-to-grape!
```

Constraints defined in clausal form can be applied to Grapes in the same way as atomic constraints, except that their negation is not defined. See below for an example.

```
(-> (newgrape) hello world vine self-or-not-hello-to-grape!)
```

```
("fca41c2d-0f44-4955-85e2-254354649eed")
```

7. Schemas

In contrast to most other graph transformation tools, ***GrapeVine*** graphs do not need to be typed. We believe that schema-less programming promotes exploration and evolution.

However, it is quite possible to define graph schemas if consistency constraints should be enforced. It is up to the user to choose how much or how little schema they desire.

Schemas are defined using the **`schema`** form

Syntax: **`(schema name [c1 ... cn])`**, where

- `name` is a name for the schema, and
- `c1` .. `cn` are constraints.

Let's consider a simple example. We define a constraint that forbids the existence of a `Hello` node with multiple outgoing `to` edges. We do so by defining an existential constraint and then using its negation:

```
(constraint multiple-to! []
       (node h:Hello)
       (node n1)
       (node n2)
       (edge :to h n1)
       (edge :to h n2))
```



Let's define a schema and call it `myschema`:

```
(schema myschema [multiple-to!-])
```

```
#'tutorial/myschema-drop
```

At this point, the schema has been defined, but it has not been attached to any graph yet. Let's define two graphs and attach the schema to one of them:

```
(def g1 (newgrape))
(def g2 (newgrape))
```

```
#'tutorial/g2
```

```
(myschema g2)
```

```
("d1a1a396-d202-4d0c-945f-8719c7a642fb")
```

The above command as attached our schema to graph `g2`.

Note that a schema is just a function on graphs (***grapes***). So, we could also just write the following:

```
(def g2 (-> (newgrape) myschema))
```

```
#'tutorial/g2
```

Now let's see the effect of the schema constraint. The following derivation should not be allowed in the schema that's attached to graph `g2` since it would create a `Hello` node connected to two `World` nodes:

```
(-> g2 hello world world)
```

```
()
```

Indeed, the above operation returns no graph (an empty ***grape***).

Of course, applying the same to our schemaless graph works without a problem:

```
(-> g1 hello world world view)
```



( )

Note: Schema constraints are "inherited" during derivation. In other words, a graph that was created as a result of a rule application has the same schema as the graph it was derived from.

7.1 Listing and Dropping Schema Constraints

We can list the schema constraints defined on a graph using the `__schema-list__` form. For example:

```
(schema-list g2)
```

```
(({:graph "5a40afa3-a573-4622-970f-3efec487ceba", :constraints ("multiple-to!-")}))
```

while...

```
(schema-list g1)
```

```
(({:graph "ad3a5296-dc66-4cac-8d85-15da46fc47e7", :constraints ()}))
```

We can drop constraints by adding a `__drop__` to the schema name. Check out the following example, which drops our schema constraints "just in time" to allow the second application of rule `world`.

```
(-> g2 hello world myschema-drop world view)
```



Note that grape `g2` still has our schema constraint attached. The above statement only dropped the constraint from the graph that was derived by applying rules `hello` and `world`.

> **Note**: Constraints (and rules) are currently not "stored procedures" in the Neo4J database. The schema is stored in the database only with the names of the constraints. This means that you need to (re)evaluate the declaration of any constraints that are referenced in a graph's schema prior to working with the graph.

7.2 Using parameterized constraints in schema definitions

Parameterized constraints can also be used in schema definitions, but their parameters must be actualized and they must be given unique names. The **`enforce`** form is used for this purpose. Consider the following example of a parameterizable version of the `multiple-to` constraint, which takes the label of node `h` as a parameter:

```
(constraint multiple-to! [lab]
        (node h:&lab)
        (node n1)
        (node n2)
        (edge :to h n1)
        (edge :to h n2))
```



The following schema uses the (negated) parameterized `multiple-to!` constraint with parameter `Hello` and names it `hello-to-one`.

```
(schema myschema [(enforce hello-to-one (multiple-to!- "Hello"))])
```

```
#'tutorial/myschema
```

7.3 Predefined schema constraints

While it is possible to define schema constraints using the general constraint form, ***GrapeVine*** comes with a list of predefined constraint types, which make schema definition more convinient.

Consider for example a constraint that restricts the allowed node types. Defining such a constraint using the above means is possible but combersome. Database schema definitions often go beyond restrictions but also defined efficient search structures like B+ trees for unique keys.

The following predefined constraint types are available:

7.3.1 Nodes and Edge Labels

We can restrict the allowed node and edge labels with the `__nodelabel__` and `__edgelabel__` forms, respectively.

Syntax: **`(nodelabels conname [l1 ... ln])`** and **`(nodelabels conname [l1 ... ln])`**, where

- `conname` is an *optional* name for the constraint, and
- `l1` .. `ln` are names of (allowed) labels

The following example defines a schema that allows only two node labels (`Hello` and `World`) and one edge label (`to`), attaches it to a graph and demonstrates its behaviour.

```
(schema myschema [
                  (nodelabels [Hello World])
                  (edgelabels   [to])
        ])

(-> (newgrape) myschema hello world view)
```



Now, the following should not be allowed, as it creates a node of type `Grape`.

```
(vine _)
```

```
()
```

7.3.2 Roles

The roles that nodes can play in relationships (as sources and targets of edges) can be restricted with the **`sourceroles`** and **`targetroles`** forms.

Syntax: **`(sourceroles conname edgelabel [l1 l2 ... ln])`** and **`(targetroles conname edgelabel [l1 l2 ... ln])`**, where

- `conname` is an *optional* name for the constraint,
- `edgelabel` is an edge label, and
- `l1` .. `ln` are labels of nodes that are allowed to play the source or target role of that edge, respectively

```
(schema myschema [
                  (sourceroles to [Hello])
                  (targetroles to [World])
                  ])

```

```
#'tutorial/myschema
```

```
(-> (newgrape) myschema hello world view)
```



Now, the following should not be allowed, since a `Grape` node cannot play the target role for a `to` edge.

```
(vine _)
```

```
()
```

7.3.3 Multiplicity

We can constrain the cardinality of relationships (edges) to have a unique target ("to one") and or unique source ("from one").

Syntax: **`(to-one conname edgelabel)`** and **`(from-one conname edgelabel)`**, where

- `conname` is an *optional* name for the constraint, and
- `edgelabel` is an edge label

```
(schema myschema [
                  (to-one to)
                  ])

```

```
#'tutorial/myschema
```

```
(-> (newgrape) myschema hello world vine view)
```



But the following does not work:

```
(-> (newgrape) myschema hello world world)
```

```
()
```

7.3.4 Attribute uniqueness

We can define attributes to be unique for all nodes of a particular label:

Syntax: **`(unique conname nodelabel attname)`**, where

- `conname` is an *optional* name for the constraint,
- `nodelabel` is a node label, and
- `attname` is an attribute name

```
(rule add-person [s]
      (create (node :Person {:SIN "'&s'"})))
```




```
(schema myschema [
                  (unique Person SIN)
                  ])


(-> (newgrape) myschema (add-person 123) (add-person 123) view)
```

```
()
```

8. Graph Queries

Graph queries are used to extract data from a graph (for further processing in the host language Clojjure) and to visualize parts of interest.

By convention and to distinguish queries from rules and constraints, names of queries should be appended with a question mark (**?**).

Like rules and constraints, queries can be applied to ***grapes***, but they return a relation for each graph in the ***grape***.

Consider the following query as an example:

```
(query hello-to? []
      (node h:Hello)
      (node x)
      (edge :to h x))
```




```
(-> (newgrape) hello world hello-to?)
```

```
(({:_g {:uid "28f18879-2e80-4360-87bd-cf703f855118"}, :h {:_fp "CC3EED4CFB833776B2B02CD1766FC47C",
:id 1937, :labels ("Hello" "__Node")}, :x {:_fp "13D744EF67DB02EC47A7D5261D2270A2", :id 1939,
:labels ("__Node" "World")}, :_oCDDXaRgSG {:tar 1939, :src 1937, :_fps
"CC3EED4CFB833776B2B02CD1766FC47C", :_fpd true, :_fpt "13D744EF67DB02EC47A7D5261D2270A2", :_fp
"5994B27BFABFBAE55A90D87A6375EDC3", :id 1940, :labels ("__Edge" "to")}}))
```

There is only one graph in the Grape that is fed to the query `hello-to?`, so the result is a set that contains only a single relation. Moreover, that relation only contains a single tuple. Extracting information of interest simply uses Clojure to navigate this set of relations. For example, the following extracts the ID of the first `h` node in the first graph:

```
(-> (newgrape) hello world hello-to? first first :h :id)
```

```
1943
```

8.1 Visualizing query results

Query results can be visualized using the `viewquery` form. Consider the following simple example of a query for "Hello" nodes and a graph that has two such nodes.

```
(-> (newgrape) hello world world world world hello-to? viewquery)
```



GrapVine comes with a built-in query `_any?` which matches any element in a graph (returning all elements as a result).

```
(-> (newgrape) hello world vine _any? viewquery)
```



Indeed, the function `view(g)` is merely a shortcut for writing `(viewquery (_any? g))`.

9. Transactions

The classical notion of transactions revolves around the ACID (atomic, consistent, isolated, durability) properties.

9.1 Durability

Since ***GrapeVine*** uses a database, all graph productions are automatically made durable. However, not all graphs may be of interest. Moreover, identifying graphs with automatically generated IDs is inconvenient.

Graphs can therefore be tagged with semantic names using the `commit` form. In the following example, we tag the resulting graph with the name "*mygraph*"

> **Note: `commit` tags an *individual* graph, that needs to be selected from a *grape*.** (In the example below, we use commit the first (only) graph in the ***grape***.)

```
(-> (newgrape) hello world vine)
(commit (first _) "mygraph")
```

```
"2687732f-9bee-4e3d-9ec8-601358036f42"
```

Once tagged, we can simply call the `graph` form to recall the graph by its semantic name, for example:

```
(-> (grape "mygraph") view)
```



The names given to graphs using the commit form must be unique. Trying to tag another graph with the same name will result in an error:

```
(-> (newgrape) hello first (commit "mygraph"))
```

```
Caught exception: org.neo4j.driver.exceptions.ClientException: Node(1651) already exists with label
`__Graph` and property `tag` = 'mygraph'
```

```
nil
```

Committed graphs can be "uncommitted" using the `uncommit` form:

```
(uncommit "mygraph")
```

```
"2687732f-9bee-4e3d-9ec8-601358036f42"
```

9.1.1 Rolling back uninteresting derivations

We may want to roll back uninteresting graph derivations, for example to free up resources or to prune the graph derivation history. We can do so by using the `rollback` form. **Warning**: calling rollback will remove *all* graphs that are not in the trace of any *committed* (tagged) graph.

The following example illustrates this behaviour. In the code block below, we create a graph history with four different branches, originating from the same empty start graph. We commit a tag for the end-result of one of these derivations.

```
(def start (newgrape))

(-> start hello world vine)
(-> start hello world vine)
(-> start hello world vine)
(-> start hello world vine)
(commit (first _) "mygraph")
(-> _ history view)
```

```
(
```




```
)
```

Calling `rollback` will remove all graphs that "do not lead up to" a committed (tagged) graph in the graph process. (see below)

```
(rollback)
(-> _ history view)
```

10. Control structures

Control structures are the primitives we use to control the application of graph transformations, i.e., to write *graph programs*. Since ***GrapeVine*** is embedded into the Clojure, all the regular Clojure control structures are available to program with graph transformations. The ***GrapeVine*** approach is to leverage these existing control structures as much as possible rather than introducing new "syntactic sugar".

We have already seen that we use simple function calls for invoking graph transformations. Calling a rule. We have also seen that the Clojure *threading macro* (->) is well-suited for *sequential composition* of graph transformations, tests and queries. Of course, we could always alternative Clojure notation (as we did before as well), e.g, we could use the classical way of nesting function calls (i.e., (f1 (f2)))

The following table provides an overview of the typical ***GrapeVine*** control structures:

Operation	*GrapeVine* construct	Comments
transformation with ***r*** (non-deterministic)	`r`	or `(r ..)` if rule has parameters
transformation with rule ***r*** (deterministic)	`r*`	or `r*(..)` if rule has parameters
distinct - remove all duplicate graphs	`distinct`	
selection of graphs that satisfy constraint ***c***	`c`	
selection of graphs that satisfy constraint ***¬c***	`c-`	
constrain graphs with invariants ***c1, . . , Cn***	`(enforce-invariant [c1, .., cn])`	transformations that would violate invariants are disallowed
assert invariants ***c1, . . , cn***	`(assert-invariant [c1, .., cn])`	invariants are monitored and violations reported
unconstrain - remove an invariants ***c1, . . , Cn*** from graphs	`(remove-invariant [c1, .., cn])`	
sequence of operations ***o1, . . on***	`(-> o1 .. on)`	all alternatives are explored concurrently
alternative of operations ***o1, . . on***	`(||o1 .. on)`	
loop of operations ***o1, . . on*** until a graph is found that satisfies constraint ***c*** - or until no further progress is possible	`(->* c o1 .. on)`	filters out duplicate graphs (using the *distinct*)
loop! of operations ***o1, . . on*** until a graph is found that satisfies constraint ***c*** - or until no further progress is possible	`(->*! c o1 .. on)`	no restriction when exploring similar graphs

10.1 An example

Consider the well-known Wolf-Goat-Cabbage (well, Wolf-Goat-Grape problem) as an example.

Below is specification of the problem. Rule `setup-ferryman` creates the starting situation. The ferryman can either take one item to the other side (rule `ferry_one_over`) or he can cross without cargo (rule `cross_empty`).

Then we have defined two graph constraints. The first one for testing the dangerious situation where a thing can be eaten (if unattended) and the second one to check whether we have solved the ridle (all on the other side).

```
(list

(rule setup-ferryman []
  (create
    (node tg:Thing {:kind "'Goat'"})
    (node tc:Thing {:kind "'Grape'"})
    (node tw:Thing {:kind "'Wolf'"})
    (node s1:Side {:name "'This side'"})
    (node s2:Side {:name "'Other side'"})
    (node f:Ferry)
    (edge :is_at tg s1)
    (edge :is_at tc s1)
    (edge :is_at tw s1)
    (edge :is_at f s1)
    ))

(rule ferry_one_over []
  (read :iso
    (node s1:Side)
    (node s2:Side)
    (node f:Ferry)
    (node t:Thing)
    (edge et:is_at t s1)
    (edge e:is_at f s1))
  (delete e et)
  (create
    (edge :is_at f s2)
    (edge :is_at t s2)))

(rule cross_empty []
  (read :iso
    (node s1:Side)
    (node s2:Side)
    (node f:Ferry)
    (edge e:is_at f s1))
  (delete e)
  (create
    (edge :is_at f s2)))

(constraint can-eat! [t1 t2]
           (node t1:Thing {:kind "'&t1'"})
           (node t2:Thing {:kind "'&t2'"})
           (node s:Side)
           (node s2:Side)
           (edge :is_at t1 s)
           (edge :is_at t2 s)
           (node f:Ferry)
           (edge :is_at f s2))

 (constraint all_on_the_other_side! []
           (node tg:Thing {:kind "'Goat'"})
           (node tc:Thing {:kind "'Grape'"})
           (node tw:Thing {:kind "'Wolf'"})
           (node s2:Side {:name "'Other side'"})
           (edge :is_at tg s2)
           (edge :is_at tc s2)
           (edge :is_at tw s2))
)

```



Below is a program that demonstrates the use of the above control structures to solve the ferryman problem:

We begin as usual with creating our start graph (grape) (line 1).

Line 2 uses are new **looping** macro. It first takes a constraint to check whether the goal has already reached for *some* graph in the ***grape***. If no graph in the ***grape*** satisfies the constraint, the loop continues. (In other words, the looping macro has an "until" semantics.).

Line 3 explores all possible ferry moves in parallel.

Finally, the (negative) constraint checks in lines 3 and 4 filter out the dangerous situations.

```
(-> (newgrape) setup-ferryman
    (->* all_on_the_other_side!
         (|| ferry_one_over* cross_empty*)
         (can-eat!- "Wolf" "Goat")
         (can-eat!- "Goat" "Grape")))
```

```
("51b2fc00-38f8-4a59-9fd4-565a5ca5b251")
```

We can confirm that a solutions was indeed found.

```
(view _)
```

```
(
```




```
)
```

It should be clear that the above program does a *breadth-first exploration* of the possible moves for the ferryman until a solution is found.

What if there is no solution? Let's modify the program and only allow the ferryman to cross *with* cargo (i.e., no empty ferries). A naive breath-first exploration would run forever, since moves with the goat are always safe but will never achieve the desired solution. The result would be an infinite graph process. The program should never terminate (disregarding resource limitations). Let's try this out:

```
(-> (newgrape) setup-ferryman
    (->* all_on_the_other_side!
         ferry_one_over*
         (can-eat!- "Wolf" "Goat")
         (can-eat!- "Goat" "Grape")))
```

```
()
```

As we can see, the program does indeed terminate and returns an empty ***grape*** (no solution). The reason for this is that the ***GrapeVine*** looping macro compares graphs in the graph history for previously seen graphs and does not further explore graphs that are similar with graphs that have been seen before.

Let's have a look at the graph history for the above program:

```
(def start (-> (newgrape) setup-ferryman))
(->* start all_on_the_other_side!
         ferry_one_over*
         (can-eat!- "Wolf" "Goat")
         (can-eat!- "Goat" "Grape"))
(-> start history steps viewsteps)

```



Indeed, there are three possible "cargo" moves of the ferry from the initial state. Two of them are "unsafe" and one of the (transporting the goat) is safe. From there, the only possible move is to transport the goat again. At that point, the resulting graph is identical to the start (up to isomorphism).

Let's look at the history of the original program again (which *has* a solution). Below we see that the breadth-first search does not explore graphs that have already been seen on the process (up to isomorphism).

```
(-> (newgrape) setup-ferryman
    (->* all_on_the_other_side!
         (|| ferry_one_over* cross_empty*)
         (can-eat!- "Wolf" "Goat")
         (can-eat!- "Goat" "Grape")))

(-> _ history view)
```



Viewing the details on the full history graph creates a big image. Rather, we are just interested in the trace that leads to the solution:

```
1 (-> _ traces steps viewsteps)
```


—setup-ferryman→
n:Thing(2188)
kind: 'Wolf'
n:Ferry(2191)
n:Thing(2187)
kind: 'Grape'
n:Side(2190)
name: 'Other side'
n:Thing(2186)
kind: 'Goat'
is_at
is_at
is_at
is_at
n:Side(2189)
name: 'This side'

If, for some reason, we do not want to limit the application of graph productions to graphs that have already been seen in the graph process (up to isomorphism), we can use the alternative looping macro **(->!*)**. The example below shows that the resulting graph process is significantly larger.

Note: this will also take longer... - be patient

```
(-> (newgrape) setup-ferryman
   (->*! all_on_the_other_side!
        (|| ferry_one_over* cross_empty*)
        (can-eat!- "Wolf" "Goat")
        (can-eat!- "Goat" "Grape")))
```

```
("43f83273-6e6c-4d55-a5d2-10ba106d1306" "dc2fdafa-a58b-4ec3-9ba3-a8e657664e60")
```

Let's see how many graphs were produced in the above computation. Do do so, we can simply "flatten" the steps involved:

```
(-> _ history steps flatten count)
```

```
216
```

We see that 216 graphs were generated in the computation that does not check for graph similarity. This compares to 27 for the computation that does (see below):

```
(-> (newgrape) setup-ferryman
   (->* all_on_the_other_side!
        (|| ferry_one_over* cross_empty*)
        (can-eat!- "Wolf" "Goat")
        (can-eat!- "Goat" "Grape")))

(-> _ history steps flatten count)
```

```
27
```

Solving the Ferryman Problem with Schema Constraints

Alternatively, we can solve the Ferryman Problem using a schema that forbids dangerous situations. We define avoidance of the two dangerous moves as schema constraints.

```
(schema safe-moves [
                    (enforce goat-is-safe  (can-eat!- "Wolf" "Goat"))
                    (enforce grape-is-safe (can-eat!- "Goat" "Grape"))
                   ])
```

```
#'tutorial/safe-moves
```

```
(-> (newgrape)
  safe-moves
  setup-ferryman
  (->* all_on_the_other_side! (|| ferry_one_over* cross_empty*)))
```

```
("ec18c2b0-6a07-4148-a186-b3e8e26c93a3")
```

11. Advanced Concepts

11.1 Paramters

Rules can be parameterized. For example, the following rule creates person nodes with a given name.

```
(rule create-person [n]
      (create (node :Person :asserts {:name "'&n'"})))
```



Formal parameters (like `n` above) must be actualized when the rule is applied. The rule application below creates two `Person` nodes with names "Flo" and "Jano", respectively.

Note that the expression `&n` is replaced with the value of the actual parameter `p` at rule execution time.

```
(-> (newgrape) (create-person "Flo") (create-person "Jano") view)
```



11.2 Negative Application Conditions (NACs)

Sometimes we want to prevent the application of rules in contexts that meet certain conditions. This can be done by specifying so-called *Negative Application Conditions* (NACs) for that rule. Consider our simple example below. Rule `world` should extend a `Hello` node with a new node `World` (and a `to` edge) only if such a node does not already exist. NACs are defined within the `read` part of the rule with the `NAC` form, as shown below.

```
(list

(rule hello []
   (create
     (node :Hello)))

(rule world []
   (read
      (node h:Hello)
      (NAC
       (node wn:World)
       (edge :to h wn )
       ))
   (create
      (node w:World)
      (edge :to h w )))
 )
```



The following statement shows that rule `world` is perfectly applicable to a graph that has only a single `Hello` node.

```
(-> (newgrape) hello world view)
```



However, trying to apply the rule again in the above result does not work:

```
(world _)
```

```
nil
```

> *Note: If we wanted to generally enforce that a `Hello` node can only ever be connected to a single `World` node, we can also define a graph constraint and declare it as an enforced invariant for rule application.*

11.3 (Positive) Application Conditions

We have already seen the use of the `asserts` clause for defining application conditions on individual nodes and edges. For other types of application conditions, it may not make sense to define them locally to a node or edge. For example, we may define a condition comparing the attribute values of multiple nodes. In that case, we can use the **`condition`** form.

The `condition` form simply takes as string that should contain a Boolean expression (in Neo4J/Cypher syntax). Rule `live-together` below shows an example:

```
(rule add-person [name address]
      (create (node :Person {:name "'&name'" :address "'&address'"})))


(rule live-together []
      (read :iso (node p1:Person)
            (node p2:Person)
            (condition "p1.address=p2.address"))
      (create (edge :together p1 p2)))
```



The following two statements give examples where the specified application condition is and isn't met, respectively.

```
(-> (newgrape) (add-person "Flo" "123 West Rd") (add-person "Jano" "321 East Ave") live-
together view)
```

```
()
```

```
(-> (newgrape) (add-person "Flo" "123 West Rd") (add-person "Jano" "123 West Rd") live-
together view)
```




```
(                                                                                )
```

11.4 Optional nodes and edges

Rules may specify certain nodes and edges as **optional**. Optional graph elements will be matched if they exist in the graph, but their presence is not mandatory. Consider the following rule `salut` as an example. It replaces a `Hello` node with a `Salut` node and optionally matches a `World` node that may be connected to it. If existent, the `World` node is preserved and reconnected to the new `Salut` node.

```
(rule salut []
      (read (node h:Hello)
            (node w:World :opt)
            (edge e:to h w :opt))
      (delete h e)
      (create (node s:Salut)
              (edge :to s w)))
```



The following example shows that rule `salut` does nto require a `World` node to exist:

```
1 (-> (newgrape) hello salut view)
```



Now, let's look at an example where a `World` node can be matched:

```
1 (-> (newgrape) hello world salut view)
```



Note: Optional graph elements will always be matched if they exist. Therefore, the use of optional graph elements in nodes does not add further non-determinism. This is demonstrated by calling the starred rule application below. It produces a single result.

```
1 (-> (newgrape) hello world salut* view)
```




```
1
```